计算概论A—实验班

函数式程序设计

Functional Programming

胡振江，张 伟

北京大学 计算机学院

2022年09～12月

Adapted from Graham's Lecture slides

第12章：Monads and More

主要知识点：

Functor、Applicative, Monad

两种提升代码抽象层次的方式

Level 1: Polymorphic Functions (over types)

```
length1 :: List a -> Int
```

Level 2: Generic Functions (over type constructors)

```
length2 :: t a -> Int
```

Functor / 函子

计算的抽象

```haskell
inc :: [Int] -> [Int]
inc []     = []
inc (n:ns) =  n+1 : inc ns
```

```haskell
sqr :: [Int] -> [Int]
sqr []     = []
sqr (n:ns) =  n^2 : sqr ns
```

抽象

```haskell
map :: (a -> b) -> [a] -> [b]
map f []     = []
map f (x:xs) = f x : map f xs
```

```haskell
inc = map (+1)  sqr = map (^1)
```

Functor

```haskell
-- Exported by Prelude
class Functor f where

    fmap :: (a -> b) -> f a -> f b

    (<$) :: b -> f a -> f b
    (<$) = fmap . const
```

```haskell
-- Exported by Prelude
const :: a -> b -> a
const x _  =  x
```

Functor

```haskell
-- Exported by Prelude
class Functor f where

    fmap :: (a -> b) -> f a -> f b

    (<$) :: a -> f b -> f a
    (<$) = fmap . const
```

```
ghci> fmap (+1) [1,2,3]
[2,3,4]
ghci> fmap (^2) [1,2,3]
[1,4,9]
```

```haskell
-- Exported by Prelude
instance Functor [] where
    -- fmap :: (a -> b) -> [a] -> [b]
    fmap = map
```

```haskell
data Maybe a = Nothing | Just a

instance Functor Maybe where
    -- fmap :: (a -> b) -> Maybe a -> Maybe b
    fmap _ Nothing  = Nothing
    fmap g (Just x) = Just (g x)
```

```
ghci> fmap (+1) (Just 3)
Just 4
ghci> fmap (+1) Nothing
Nothing
ghci> fmap not (Just False)
Just True
```

```haskell
data Tree a = Leaf a | Node (Tree a) (Tree a)
              deriving (Show)

instance Functor Tree where
    -- fmap :: (a -> b) -> Tree a -> Tree b
    fmap g (Leaf x)   = Leaf $ g x
    fmap g (Node l r) = Node (fmap g l) (fmap g r)
```

```
ghci> fmap length (Leaf "abc")
Leaf 3
ghci> fmap even $ Node (Leaf 1) (Leaf 2)
Node (Leaf False) (Leaf True)
```

```haskell
instance Functor IO where
    -- fmap :: (a -> b) -> IO a -> IO b
    fmap g mx = do x <- mx
                   return $ g x
```

```
ghci> fmap show $ return True
"True"
```

Generic Function Definition

```
inc :: Functor f => f Int -> f Int
inc = fmap (+1)
```

```
ghci> inc $ Just 1
Just 2
ghci> inc [1,2,3,4,5]
[2,3,4,5,6]
ghci> inc $ Node (Leaf 1) (Leaf 2)
Node (Leaf 2) (Leaf 3)
```

Functor Laws

① `fmap id = id`

② `fmap (f . g) = fmap f . fmap g`

✤ **For any parameterized type in Haskell, there is *at most one* function fmap that satisfies the required laws.**

- That is, if it is possible to make a given parameterized type into a functor, there is only one way to achieve this.
- Hence, the instances that we defined for lists, Maybe, Tree and IO were all uniquely determined.

<$> : An infix synonym for fmap

```haskell
-- Exported by Prelude
(<$>) :: Functor f => (a -> b) -> f a -> f b
(<$>) = fmap
```

The name of this operator is an allusion to `$`. Note the similarities between their types:

```
  ($)  ::                 (a -> b) ->   a ->   b
(<$>) :: Functor f => (a -> b) -> f a -> f b
```

Whereas `$` is function application, `<$>` is function application lifted over a `Functor`.

Applicative

Applicative Functor

如何定义一个一般性的fmap

```
fmap0 :: a -> f a
```

```
fmap1 :: (a -> b) -> f a -> f b
```

```
fmap2 :: (a -> b -> c) -> f a -> f b -> fc
```

```
fmap3 :: (a -> b -> c -> d) -> f a -> f b -> fc -> f d
```

⋮

两个基本函数

```
pure :: a -> f a
```

```
(<*>) :: f (a -> b) -> f a -> f b
```

```
pure :: a -> f a
```

```
(<*>) :: f (a -> b) -> f a -> f b
```

```
fmap0 :: a -> f a
```

```
fmap0 = pure
```

```
fmap1 :: (a -> b) -> f a -> f b
```

```
fmap1 g x = pure g <*> x
```

```
fmap1 g x = fmap g x = g <$> x
```

```
fmap2 :: (a -> b -> c) -> f a -> f b -> fc
```

```
fmap2 g x y = pure g <*> x <*> y = g <$> x <*> y
```

```
fmap3 :: (a -> b -> c -> d) -> f a -> f b -> fc -> f d
```

```
fmap3 g x y z = pure g <*> x <*> y <*> z = g <$> x <*> y <*> z
```

Applicative Functor

Applicative Functor: 一个简化版本

```haskell
class Functor f => Applicative f where

    -- Lift a value
    pure :: a -> f a

    -- Sequential application.
    (<*>) :: f (a -> b) -> f a -> f b
```

Applicative Functor: 一个简化版本

```haskell
class Functor f => Applicative f where
    -- Lift a value
     pure :: a -> f a
    -- Sequential application.
     (<*>) :: f (a -> b) -> f a -> f b
```

声明 Maybe为Applicative的一个实例

```haskell
instance Applicative Maybe where
    -- pure :: a -> Maybe a
    pure = Just

    -- (<*>) :: Maybe (a -> b) -> Maybe a -> Maybe b
    Nothing  <*> _    = Nothing
    (Just g) <*> mx   = g <$> mx
```

Applicative Functor: –

```haskell
class Functor f => Appli
    -- Lift a value
     pure :: a -> f a
    -- Sequential applica
     (<*>) :: f (a -> b)
```

```
ghci> pure (+1) <*> Just 1
Just 2
ghci> pure (+) <*> Just 1 <*> Just 2
Just 3
ghci> pure (+) <*> Nothing <*> Just 2
Nothing
ghci> Nothing <*> Just 1
Nothing
```

声明 Maybe为Applicative的一个实例

```haskell
instance Applicative Maybe where
    -- pure :: a -> Maybe a
    pure = Just

    -- (<*>) :: Maybe (a -> b) -> Maybe a -> Maybe b
    Nothing  <*> _    = Nothing
    (Just g) <*> mx   = g <$> mx
```

声明 [ ] 为Applicative的一个实例

```haskell
instance Applicative [] where
    -- pure :: a -> [a]
    pure x = [x]

    -- (<*>) :: [a -> b] -> [a] -> [b]
    gs <*> xs = [g x | g <- gs, x <- xs]
```

```
ghci> pure (+1) <*> [1,2,3]
[2,3,4]
ghci> pure (+) <*> [1] <*> [2]
[3]
ghci> pure (*) <*> [1,2] <*> [3,4]
[3,4,6,8]
```

声明 IO 为Applicative的一个实例

```haskell
instance Applicative IO where
  -- pure :: a -> IO a
  pure = return

  -- (<*>) :: IO (a -> b) -> IO a -> IO b
  mg <*> mx = do {g <- mg; x <- mx; return (g x)}
```

```haskell
getChars :: Int -> IO String
getChars 0 = return []
getChars n = pure (:) <*> getChar <*> getChars (n-1)
```

Generic Function Definition

```
sequenceA :: Applicative f => [f a] -> f [a]
sequenceA []     = pure []
sequenceA (x:xs) = pure (:) <*> x <*> sequenceA xs
```

```
ghci> sequenceA [Just 1, Just 2, Just 3]
Just [1,2,3]
ghci> sequenceA [Just 1, Nothing, Just 3]
Nothing
ghci> sequenceA [[1,2,3], [4,5,6], [7,8,9]]
[[1,4,7],[1,4,8],[1,4,9],[1,5,7],[1,5,8],[1,5,9],[1,6,7],[1,6,8],[1,6,9],
[2,4,7],[2,4,8],[2,4,9],[2,5,7],[2,5,8],[2,5,9],[2,6,7],[2,6,8],[2,6,9],
[3,4,7],[3,4,8],[3,4,9],[3,5,7],[3,5,8],[3,5,9],[3,6,7],[3,6,8],[3,6,9]]
```

Applicative Laws

① `pure id <*> x = x`

② `pure (g x) = pure g <*> pure x`

③ `x <*> pure y = pure (\g -> g y) <*> x`

④ `x <*> (y <*> z) = (pure (.) <*> x <*> y) <*> z`

Applicative Laws: 类型分析

```
pure id <*> x = x
```

```
a -> a
```

```
f a
```

```
pure (g x) = pure g <*> pure x
```

```
f b
```

```
a -> b
```

```
a
```

Applicative Laws: 类型分析

```
x <*> pure y = pure (\g -> g y) <*> x
```




```
x <*> (y <*> z) = (pure (.) <*> x <*> y) <*> z
```

Applicative Laws: 类型分析



Monad

一个小问题：异常处理

```haskell
data Expr = Val Int | Div Expr Expr

eval :: Expr -> Int
eval (Val n)   = n
eval (Div x y) = eval x `div` eval y
```

```
ghci> eval $ Div (Val 1) (Val 0)
*** Exception: divide by zero
```

解決方法1

```haskell
safediv :: Int -> Int -> Maybe Int
safediv _ 0 = Nothing
safediv n m = Just (n `div` m)
```

```haskell
eval :: Expr -> Maybe Int
eval (Val n)   = Just n
eval (Div x y) = case eval x of
                   Nothing -> Nothing
                   Just n  -> case eval y of
                     Nothing -> Nothing
                     Just m  -> safediv n m
```

稍显繁杂

解决方法2

```haskell
safediv :: Int -> Int -> Maybe Int
safediv _ 0 = Nothing
safediv n m = Just (n `div` m)
```

```haskell
eval :: Expr -> Maybe Int
eval (Val n)   = pure n
eval (Div x y) = pure safediv <*> eval x <*> eval y
```

类型错误

type: Maybe (Maybe Int)

解決方法2

```haskell
safediv :: Int -> Int -> Maybe Int
safediv _ 0 = Nothing
safediv n m = Just (n `div` m)
```

```haskell
eval :: Expr -> Maybe Int
eval (Val n)   = pure n
eval (Div x y) = case pure safediv <*> eval x <*> eval y of
                   Just r -> r
                   Nothing -> Nothing
```

Maybe (Maybe Int)



還是不够簡洁

解决方法3：引入一个新的操作 bind

```haskell
(>>=) :: Maybe a -> (a -> Maybe b) -> Maybe b
mx >>= f = case mx of
             Nothing -> Nothing
             Just x  -> f x
```

```haskell
eval :: Expr -> Maybe Int
eval (Val n)    = Just n
eval (Div x y)  = eval x >>= (\n -> (eval y >>= (\m -> safediv n m)))
```



解决方法3：引入一个新的操作 bind

```haskell
(>>=) :: Maybe a -> (a -> Maybe b) -> Maybe b
mx >>= f = case mx of
             Nothing -> Nothing
             Just x  -> f x
```

```haskell
eval :: Expr -> Maybe Int
eval (Val n)    = Just n
eval (Div x y)  = eval x >>= (\n -> (eval y >>= (\m -> safediv n m)))
```

先耍一点朝三暮四的小把戏

```haskell
eval :: Expr -> Maybe Int
eval (Val n)    = Just n
eval (Div x y)  = eval x >>= \n ->
                  eval y >>= \m ->
                  safediv n m
```

=

再撒一点扑朔迷离的语法糖

```haskell
eval :: Expr -> Maybe Int
eval (Val n)    = Just n
eval (Div x y)  = do n <- eval x
                     m <- eval y
                     safediv n m
```

Monad

```haskell
{- The Monad class defines the basic operations over a monad,
   a concept from a branch of mathematics known as "category theory".
   From the perspective of a Haskell programmer, however,
   it is best to think of a monad as an abstract datatype of actions.
   The do expressions provide a convenient syntax for writing monadic expressions.-}
class Applicative m => Monad m where

    -- Inject a value into the monadic type.
    return :: a -> m a
    return = pure

    -- Sequentially compose two actions,
    -- passing any value produced by the first as an argument to the second.
    (>>=) :: m a -> (a -> m b) -> m b

    -- Sequentially compose two actions, discarding any value produced by the first,
    -- like sequencing operators (such as the semicolon) in imperative languages.
    (>>) :: m a -> m b -> m b
    m >> k = m >>= \_ -> k
```

```haskell
a >>= f
```

=

```haskell
do v <- a
   f v
```

```haskell
a >> b
```

=

```haskell
do a
   b
```

=

```haskell
do v <- a
   b
```

声明 Maybe 为Monad的一个实例

```haskell
class Applicative m => Monad m where
    return :: a -> m a
    return = pure

    (>>=) :: m a -> (a -> m b) -> m b

    (>>) :: m a -> m b -> m b
    m >> k = m >>= \_ -> k
```

```haskell
instance Monad Maybe where
    -- (>>=) :: Maybe a -> (a -> Maybe b) -> Maybe b
    Nothing  >>= _ = Nothing
    (Just x) >>= f = f x
```

声明 [ ] 为Monad的一个实例

```haskell
class Applicative m => Monad m where
    return :: a -> m a
    return = pure

    (>>=) :: m a -> (a -> m b) -> m b

    (>>) :: m a -> m b -> m b
    m >> k = m >>= \_ -> k
```

```haskell
instance Monad [] where
    -- (>>=) :: [a] -> (a -> [b]) -> [b]
    xs >>= f = [y | x <- xs, y <- f x]
```

The State Monad

✤ 问题：如何用函数描述状态的变化

- 状态：一种数据类型
  - `type State = Int`
  - 仅仅是一个示例；需根据具体问题确定状态的类型
- 状态变换器
  - `type ST = State -> State`
- 带有结果的状态变换器
  - `type ST a = State -> (a, State)`



用type声明的ST不是一个Type Constructor，无法将ST声明为Functor/Applicative/Monad的实例

用 newtype 定义 ST

```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```

将 ST 声明为 Functor 的实例

```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```




```haskell
instance Functor ST where
  -- fmap :: (a -> b) -> ST a -> ST b
  fmap g st = S
```

将 ST 声明为 Functor 的实例

```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```




```haskell
instance Functor ST where
  -- fmap :: (a -> b) -> ST a -> ST b
  fmap g st = S $ \s -> let (x, s') = app st s in (g x, s')
```

将 ST 声明为 Applicative 的实例

```haskell
instance Applicative ST where
    -- pure :: a -> ST a
    pure x = S [redacted]

    -- (<*>) :: ST (a -> b) -> ST a -> ST b
    stf <*> stx = S [redacted]
```




```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```

将 ST 声明为 Applicative 的实例

```haskell
instance Applicative ST where
    -- pure :: a -> ST a
    pure x = S $ \s -> (x,s)

    -- (<*>) :: ST (a -> b) -> ST a -> ST b
    stf <*> stx = S
```






```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```

将 ST 声明为 Applicative 的实例

```haskell
instance Applicative ST where
    -- pure :: a -> ST a
    pure x = S $ \s -> (x,s)

    -- (<*>) :: ST (a -> b) -> ST a -> ST b
    stf <*> stx = S $ \s -> let (f, s' ) = app stf s
                               (x, s'') = app stx s'
                           in  (f x, s''))
```






```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```

将 ST 声明为 Monad 的实例

```haskell
instance Monad ST where
    -- (>>=) :: ST a -> (a -> ST b) -> ST b
    st >>= f = S
```




```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```

将 ST 声明为 Monad 的实例

```haskell
instance Monad ST where
    -- (>>=) :: ST a -> (a -> ST b) -> ST b
    st >>= f = S $ \s -> let (x,s') = app st s
                         in app (f x) s')
```




```haskell
newtype ST a = S (State -> (a, State))

app :: ST a -> State -> (a, State)
app (S f) s = f s
```

The State Monad

这几张幻灯片讲的挺好的

下次不要再讲了

感觉讲了一些无用的废话

在我第一次看到State Monad时
内心的想法其实也和你们差不多

The State Monad 之 应用示例：树的重新标注

```haskell
data Tree a = Leaf a | Node (Tree a) (Tree a)
              deriving Show

tree :: Tree Char
tree = Node (Node (Leaf 'a') (Leaf 'b')) (Leaf 'c')
```

- Consider the problem of defining a function that relabels each leaf in such a tree with a unique or fresh integer.

```
ghci> relabel tree
Node (Node (Leaf 0) (Leaf 1)) (Leaf 2)
```

树的重新标注 之 方法一：朴实无华~隐入尘烟

```haskell
rlabel :: Tree a -> Int -> (Tree Int, Int)
rlabel (Leaf _  ) n = (Leaf n, n+1)
rlabel (Node l r) n = (Node l' r', n'')
                  where (l', n' ) = rlabel l n
                        (r', n'') = rlabel r n'

relabel :: Tree a -> Tree Int
relabel t = fst (rlabel t 0)
```

缺点：rlabel 的定义中需要显式维护中间状态

```
ghci> relabel tree
Node (Node (Leaf 0) (Leaf 1)) (Leaf 2)
```

树的重新标注 之 方法二：Applicative

```haskell
fresh :: ST Int
fresh = S $ \n -> (n, n+1)

alabel :: Tree a -> ST (Tree Int)
alabel (Leaf _)   = Leaf <$> fresh
alabel (Node l r) = Node <$> alabel l <*> alabel r

relabel' :: Tree a -> Tree Int
relabel' t = fst $ app (alabel t) 0
```





<$> = `fmap`

or

g <$> x = pure g <*> x

我时常在想，这些东西是永恒的吗?

如果是，它们栖身何处，以至可以被人类发现并表达



树的重新标注 之 方法三： Monad

```
mlabel :: Tree a -> ST (Tree Int)
mlabel (Leaf _)   = fresh >>= \n -> return $ Leaf n
mlabel (Node l r) = mlabel l >>= \l' ->
                    mlabel r >>= \r' -> return $ Node l' r'

relabel'' :: Tree a -> Tree Int
relabel''  t = fst $ app (mlabel t) 0
```




```
mlabel (Leaf _  ) = do n <- fresh
                       return (Leaf n)
mlabel (Node l r) = do l' <- mlabel l
                       r' <- mlabel r
                       return $ Node l' r'
```

使用 do 改写 mlabel

Monad Laws

Left identity	`return a >>= h = h a`
Right identity	`m >>= return = m`
Associativity	`(m >>= g) >>= h = m >>= (\x -> g x >>= h)`
	`( m >>= \x -> g x) >>= h = m >>= (\x -> g x >>= h)`

```
class Applicative m => Monad m where
    return :: a -> m a
    return = pure

    (>>=) :: m a -> (a -> m b) -> m b

    (>>) :: m a -> m b -> m b
    m >> k = m >>= \_ -> k
```

Monad Laws: Another Form

```haskell
class Applicative m => Monad m where
    return :: a -> m a
    return = pure
    (>>=) :: m a -> (a -> m b) -> m b
    ...
```

```haskell
-- defined in Control.Monad
(>=>) :: Monad m => (a -> m b) -> (b -> m c) -> (a -> m c)
f >=> g     = \x -> f x >>= g
```

Left identity	return a	>>=	h	=	h a
	(\x -> return x	>>=	h) a	=	h a
	(return	>=>	h) a	=	h a
	return	>=>	h	=	h

看！是不是 Left identity

Monad Laws: Another Form

```
class Applicative m => Monad m where
    return :: a -> m a
    return = pure
    (>>=) :: m a -> (a -> m b) -> m b
    ...
```

```
-- defined in Control.Monad
(>=>) :: Monad m => (a -> m b) -> (b -> m c) -> (a -> m c)
f >=> g     = \x -> f x >>= g
```

Right identity	return a	>>=	h	=	h a
	(\x -> return x	>>=	h) a	=	h a
	(return	>=>	h) a	=	h a
	return	>=>	h	=	h

Monad Laws: Another Form

```
class Applicative m => Monad m where
    return :: a -> m a
    return = pure
    (>>=) :: m a -> (a -> m b) -> m b
    ...
```

```
-- defined in Control.Monad
(>=>) :: Monad m => (a -> m b) -> (b -> m c) -> a -> m c
f >=> g     = \x -> f x >>= g
```

Left identity	`return a >>= h`	=	`h a`
	`(\x -> return x >>= h) a`	=	`h a`
Right identity	`m >>= return`	=	`m`
Associativity	`(m >>= g) >>= h`	=	`m >>= (\x -> g x >>= h)`
	`( m >>= \x -> g x) >>= h`	=	`m >>= (\x -> g x >>= h)`

Monad Laws: Another Form

```
class Applicative m => Monad m where
    return :: a -> m a
    return = pure
    (>>=) :: m a -> (a -> m b) -> m b
    ...
```

```
-- defined in Control.Monad
(>=>) :: Monad m => (a -> m b) -> (b -> m c) -> a -> m c
f >=> g     = \x -> f x >>= g
```

Left identity	`return a >>= h`	=	`h a`
	`(\x -> return x >>= h) a`	=	`h a`
Right identity	`m >>= return`	=	`m`
Associativity	`(m >>= g) >>= h`	=	`m >>= (\x -> g x >>= h)`
	`( m >>= \x -> g x) >>= h`	=	`m >>= (\x -> g x >>= h)`

课堂练习 1

- Define an instance of the Functor class for the following type of binary trees that have data in their nodes:

```
data Tree a = Leaf | Node (Tree a) a (Tree a) deriving (Show)
```

```
instance Functor Tree where
    -- fmap :: (a -> b) -> Tree a -> Tree b
```

课堂练习 1

- Define an instance of the Functor class for the following type of binary trees that have data in their nodes:

```
data Tree a = Leaf | Node (Tree a) a (Tree a) deriving (Show)
```

```
instance Functor Tree where
    -- fmap :: (a -> b) -> Tree a -> Tree b
    fmap g Leaf = Leaf
    fmap g (Node l x r) = Node (fmap g l) (g x) (fmap g r)
```

课堂练习 2

✤ Complete the following instance declaration to make the partially-applied function type `(->) a` into a functor:

```
instance Functor ((->) a) where
  -- fmap :: (a -> b) -> f a -> f b
```

课堂练习 2

✤ Complete the following instance declaration to make the partially-applied function type `(->) a` into a functor:

```
instance Functor ((->) a) where
  -- fmap :: (a -> b) -> f a -> f b
  -- fmap :: (b -> c) -> f b -> f c
```

课堂练习 2

✤ Complete the following instance declaration to make the partially-applied function type `(->) a` into a functor:

```
instance Functor ((->) a) where
  -- fmap :: (a -> b) -> f a -> f b
  -- fmap :: (b -> c) -> f b -> f c
  -- fmap :: (b -> c) -> (->) a b -> (->) a c
```

课堂练习 2

✤ Complete the following instance declaration to make the partially-applied function type `(->) a` into a functor:

```
instance Functor ((->) a) where
  -- fmap :: (a -> b) -> f a -> f b
  -- fmap :: (b -> c) -> f b -> f c
  -- fmap :: (b -> c) -> (->) a b -> (->) a c
  -- fmap :: (b -> c) -> (a -> b) -> (a -> c)
```

如果一个东西可以被定义为Functor的实例，那么，只有一种fmap的定义方式

课堂练习 2

✤ Complete the following instance declaration to make the partially-applied function type `(->) a` into a functor:

```
instance Functor ((->) a) where
  -- fmap :: (a -> b) -> f a -> f b
  -- fmap :: (b -> c) -> f b -> f c
  -- fmap :: (b -> c) -> (->) a b -> (->) a c
  -- fmap :: (b -> c) -> (a -> b) -> (a -> c)
  fmap = (.)
```

如果一个东西可以被定义为Functor的实例，那么，只有一种fmap的定义方式

课堂练习 3

✤ Define an instance of the Applicative class for the type `(->) a`

```haskell
instance Applicative ((->) a) where
  -- pure :: a -> f a
  -- (<*>) :: f (a -> b) -> f a -> f b
```

课堂练习 3

✤ Define an instance of the Applicative class for the type `(->) a`

```haskell
instance Applicative ((->) a) where
  -- pure :: a -> f a
  -- pure :: b -> f b
  
  
  
  -- (<*>) :: f (a -> b) -> f a -> f b
  -- (<*>) :: f (b -> c) -> f b -> f c
  
  
  
```

课堂练习 3

✤ Define an instance of the Applicative class for the type `(->) a`

```haskell
instance Applicative ((->) a) where
  -- pure :: a -> f a
  -- pure :: b -> f b
  -- pure :: b -> a -> b

  -- (<*>) :: f (a -> b) -> f a -> f b
  -- (<*>) :: f (b -> c) -> f b -> f c
  -- (<*>) :: (a -> b -> c) -> (a -> b) -> (a -> c)
```

课堂练习 3

✤ Define an instance of the Applicative class for the type `(->) a`

```haskell
instance Applicative ((->) a) where
  -- pure :: a -> f a
  -- pure :: b -> f b
  -- pure :: b -> a -> b
  pure = const

  -- (<*>) :: f (a -> b) -> f a -> f b
  -- (<*>) :: f (b -> c) -> f b -> f c
  -- (<*>) :: (a -> b -> c) -> (a -> b) -> (a -> c)
```

课堂练习 3

✤ Define an instance of the Applicative class for the type `(->) a`

```haskell
instance Applicative ((->) a) where
  -- pure :: a -> f a
  -- pure :: b -> f b
  -- pure :: b -> a -> b
  pure = const

  -- (<*>) :: f (a -> b) -> f a -> f b
  -- (<*>) :: f (b -> c) -> f b -> f c
  -- (<*>) :: (a -> b -> c) -> (a -> b) -> (a -> c)
  g <*> h = \x -> g x $ h x
```

作业

12-1 Define an instance of the Monad class for the type `(->) a`.

12-2 Given the following type of expressions

```
data Expr a = Var a | Val Int | Add (Expr a) (Expr a)
              deriving Show
```

that contain variables of some type a, show how to make this type into instances of the Functor, Applicative and Monad classes. With the aid of an example, explain what the >>= operator for this type does.

Adapted from Graham's Lecture slides

第10章：Monads and More

就到这里吧